# $Id: PKGBUILD 163956 2012-07-22 18:33:26Z schiv $
# Maintainer: Ray Rashif <schiv@archlinux.org>
# Contributor: Pierre Schmitz <pierre@archlinux.de>
# Contributor: Paul Mattal <paul@archlinux.org>

pkgname=ddrescue
pkgver=1.16
pkgrel=2
pkgdesc="GNU data recovery tool"
arch=('i686' 'x86_64')
url="http://www.gnu.org/software/ddrescue/ddrescue.html"
license=('GPL3')
depends=('gcc-libs')
install=$pkgname.install
source=("http://ftp.gnu.org/gnu/$pkgname/$pkgname-$pkgver.tar.gz"
        "http://ftp.gnu.org/gnu/$pkgname/$pkgname-$pkgver.tar.gz.sig")
md5sums=('57b67407e882c6418531d48a2f20d16b'
         'bf072280587665d82829be15eb6fc9ad')

build() {
  cd "$srcdir"/$pkgname-$pkgver
  ./configure --prefix=/usr
  make
}

package() {
  cd "$srcdir"/$pkgname-$pkgver
  make DESTDIR="$pkgdir" install install-man
  rm "$pkgdir"/usr/share/info/dir
}
